

Динамика изменения количества природоохранных зон в России и мире

Andrew Krizhanovsky^a, Mihail Stepanov^b

^a*Institute of Applied Mathematical Research of the Karelian Research Centre of the Russian Academy of Sciences*

^b*Petrozavodsk State University*

Abstract

Аннотация

Статья посвящена исследованию объекта Викиданных "природоохранная зона". С помощью SPARQL-запросов были решены следующие задачи: выведен список всех экземпляров объекта "природоохранная зона" выведен список всех экземпляров объекта "природоохранная зона" находящихся на территории России, построена столбчатая диаграмма, отображающая динамику изменения количества новых природоохранных зон в России и в мире по десятилетиям, построена пузырьковая диаграмма, отображающая соотношение количества природоохранных зон в разных странах.

Также был произведен анализ полноты Викиданных, дополнена информация у 100 экземпляров объекта "природоохранная зона".

Примечание

Статья распространяется по лицензии Creative Commons Attribution-ShareAlike. Материалы этой статьи использованы в главе курса Викиверситета "Программирование Викиданных"[1]. Иллюстрации загружены на Викисклад. Над статьёй в 2017 году работали Крижановский А. А., Степанов М. Я.

Экземпляры объекта "Природоохранная зона"

- Объекты: природоохранная зона (Q473972).
- Свойства: экземпляр(P31), подкласс (P279).

Построим список всех природоохранных зон. См. листинг 1.

```
1 #List of 'instances of' "protected area"
2 SELECT ?area ?areaLabel
3 WHERE
4 {
5   ?area (wdt:P31/wdt:P279*) wd:Q473972 .
6   SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
7 }
```

Листинг 1: Список всех природоохранных зон

SPARQL-запрос, 32639 записей.

Наиболее полными и проработанными природоохранными зонами на Викиданных являются: Цзючжайгоу, Кракатау, Риф Кингмен.

Почти пустыми и малоинформативными природоохранными зонами оказались: Шульган-Таш (заповедник), Оренбургский заповедник, Аджлаун Форест (заказник).

Природоохранные зоны России

Выведем список природоохранных зон, находящихся на территории России (листинг 2).

- Объекты: природоохранная зона (Q473972), Россия (Q159).
- Свойства: экземпляр (P31), государство (P17), подкласс (P279).

```
1 #added 2017-03
2 SELECT ?area ?areaLabel           #List of 'instances' ...
3 WHERE
4 {
5   ?area (wdt:P31/wdt:P279*) wd:Q473972.   # ... of "protected area" ...
6   ?area wdt:P17 wd:Q159.                 # ... belonging to "Russia"
7   SERVICE wikibase:label { bd:serviceParam wikibase:language "ru" }
8 }
```

Листинг 2: Список всех природоохранных зон России

SPARQL-запрос, 819 записей.

Природоохранные зоны по дате основания

Построим гистограмму, отображающую количество оснований (созданий) природоохранных зон по десятилетиям. См. листинги 3 и 4.

- Объекты: природоохранная зона (Q473972), Россия (Q159), СССР(Q15180), Российская империя (Q34266).
- Свойства: экземпляр (P31), государство (P17), дата основания/создания (P571).

По всему миру

```
1 #added 2017-03
2 #defaultView: BarChart
3 SELECT (str(?decade) AS ?decade) (COUNT(?inception) AS ?quantity) # number of
4   "inceptions" ...
5 WHERE {
6   ?item wdt:P31 wd:Q473972. # ... of "protected areas" ...
7   ?item wdt:P571 ?inception.
8   BIND( YEAR(?inception) as ?year )
9   BIND( FLOOR(?year/10)*10 as ?decade ) # ... per each 10 years
```

```

10 }
11 GROUP BY ?decade
12 ORDER BY ?decade

```

Листинг 3: Основания всех природоохранных зон по десятилетиям

SPARQL-запрос, 12 записей.

На рис. 1 представлен результат работы скрипта (листинг 3).




Рис. 1: Столбчатая диаграмма, отображающая количество созданных природоохранных зон в мире на каждую декаду (10 лет), согласно заполненному свойству “дата основания” (inception) у объектов. Отображены не все декады, а только те, в которые был открыт хотя бы один заповедник.

Первой созданной в мире природоохранной зоной, согласно Викиданным, была Большая мечеть в Самарре, 848-852 годы.

По России

```

1 #added 2017-03
2 #defaultView:BarChart
3 # Do bar chart as result representation ...
4 SELECT (str(?decade) AS ?decade) (COUNT(?inception) AS ?areas) # ... of the
   number of "inceptions" ...
5 WHERE {
6   ?item wdt:P31 wd:Q473972. # ... of "protected areas" ...
7   ?item wdt:P571 ?inception.
8
9   { ?item wdt:P17 wd:Q159 } UNION # ... belonging to "Russia", ...
10  { ?item wdt:P17 wd:Q15180 } UNION # ... "Soviet Union" ...
11  { ?item wdt:P17 wd:Q34266 } # ... and "Russian Empire" ...
12
13  BIND(YEAR(?inception) as ?year )
14  BIND( FLOOR(?year/10)*10 as ?decade ) # ... per each 10 years
15 }
16 GROUP BY ?decade

```

Листинг 4: Основания природоохранных зон по десятилетиям в России

SPARQL-запрос, 7 записей.

На рис. 2 представлен результат работы скрипта (листинг 4).




Рис. 2: Столбчатая диаграмма, отображающая подекадно количество новых природоохранных зон на территории России, согласно заполненному свойству “дата основания” у объектов. Отображены не все декады, а только те, в которые был открыт хотя бы один заповедник.

Первой созданной в России природоохранной зоной, согласно Викиданным, была Кедровая Падь, 1916 год.

Сравнение и анализ динамики изменения количества новых природоохранных зон в России и во всем мире

Сравним рис. 1 и рис. 2, укажем на их сходства:

- Сокращение количества новых зон в 1940-х (три зоны) и 1960-х (12 зон) годах в мире и полное отсутствие в России. Вероятно, что причиной этому были Вторая мировая война и экстенсивное развитие экономики в 1960-х годах соответственно.
- Активный рост числа заповедных территорий с начала 1970-х годов. Максимальные показатели в 1990-х годах: 12 новых зон в России, 185 в мире.
- Резкое падение числа новых природоохранных территорий после 2010 года (16 зон в мире, одна - в России). Вероятнее всего ситуация такова из-за того, что человечество уже взяло под контроль большую часть требующих того природных зон. Также, возможно, данные в Википедии могут отставать от действительности и будут введены позже.

Различия:

- Резкий рост числа новых зон в 1950-х годах в мире (46 новых зон), снижение роста в России (одна новая зона - национальный парк Шульган-Таш).

Природоохранные зоны в разных странах

Покажем соотношение числа природоохранных зон в разных странах мира с помощью пузырьковой диаграммы (листинг 5).

- Объекты: природоохранная зона (Q473972).
- Свойства: экземпляр (P31), государство (P17), подкласс (P279).

```
1 #defaultView:BubbleChart
2 SELECT ?area ?country (count(*) as ?count) # number of ...
3 WHERE
4 {
5   ?pid (wdt:P31/wdt:P279*) wd:Q473972. # ... "protected areas" ...
6   ?pid wdt:P17 ?area. # ... in different countries of the world
7   OPTIONAL {
8     ?area rdfs:label ?country
9     filter (lang(?country) = "en")
10  }
11 }
12 GROUP BY ?area ?country
13 ORDER BY DESC(?count) ASC(?country)
```

Листинг 5: Соотношение числа природоохранных зон в разных странах

SPARQL-запрос, 184 записи.

В результате выполнения скрипта (листинг 5) получены 184 страны с заповедниками (рис. 3). Лидирующие позиции занимают Соединенные Штаты Америки (5075 зон), Германия (4921 зона), Чехия (4111 зон). Россия располагается на девятом месте (819 зон) рядом с Румынией и Словакией.

Полнота Викиданных

Незаполненное свойство "дата основания"

Согласно данным статьи "Природоохранная зона" Русской Википедии в мире существует 147 тысяч охраняемых территорий. В качестве авторитетного источника приведена статья В. В. Дёжкина "Особо охраняемые природные территории мира" однако ссылка является нерабочей. Эта статья была найдена в сети Интернет, но она содержит устаревшие данные за 2003 год (102 тысячи охраняемых территорий) [2].

Согласно данным статьи "Protected area" Английской Википедии в мире существует 161 тысяча охраняемых территорий. В качестве авторитетного источника приведен интернет-сервис Protected Planet, управляющийся Всемирным центром мониторинга охраны окружающей среды. Этот сайт является самой полной базой данных природоохранных зон, однако на данный момент позволяет получать глобальную статистику только по отдельно взятой стране. Например, по данным сервиса на территории России располагается 11,251 природоохранная зона [3].

Также, согласно данным с сайта международного форума IUCN World Parks Congress, в мире насчитывается около двухсот тысяч охраняемых территорий [4].




Рис. 3: Пузырьковая диаграмма, показывающая соотношение количества природоохранных зон в разных странах.

С помощью первого SPARQL-запроса (листинг 1) мы получаем 32,639 результатов по всему миру, потому что о большинстве природоохранных зон нет записей в Викиданных, вероятно, они имеют низкую значимость. В основном это большинство составляют природоохранные зоны регионального и областного значения. Например, по данным статьи "Особо охраняемые природные территории России" Русской Википедии федерального значения имеют только 299 особо охраняемых природных территорий.

Если обращаться к SPARQL-запросам, отображающим природоохранные зоны по дате основания, то здесь результатов еще меньше. Например, по России Викиданные содержат всего 25 зон. Это связано с тем, что у большинства объектов не заполнено свойство "дата основания" (inception).

Выведем эти объекты при помощи скрипта (листинг 6).

1 #added 2017-03

```

2 SELECT ?area ?areaLabel ?inception #List of 'instances' ...
3 WHERE
4 {
5   ?area wdt:P31 wd:Q473972. # ... of "protected area" ...
6   ?area wdt:P17 wd:Q159. # ... belonging to "Russia"
7
8   FILTER NOT EXISTS { ?area wdt:P571 [] } #... with unfilled property "
9   inception"
10  SERVICE wikibase:label { bd:serviceParam wikibase:language "ru" }

```

Листинг 6: Природоохранные зоны России с незаполненным свойством "дата основания"

SPARQL-запрос, 262 результата.

В результате выполнения скрипта (листинг 6) получен список экземпляров объекта "природоохранная зона принадлежащих к России, с незаполненным свойством "дата основания"(до 20.03.2017).

Заполнение свойства "дата основания" у 100 объектов




Рис. 4: Столбчатая диаграмма, отображающая подекадно количество основанных природоохранных зон на территории России, согласно заполненному свойству "дата основания" у объектов, после заполнения 100 объектов от 20.03.2017. Отображены не все декады, а только те, в которые был открыт хотя бы один заповедник.

На 19 марта 2017 года Викиданные содержали 25 природоохранных зон по России с заполненным свойством "дата основания"(см. листинг 4, рис. 2).

На сегодняшний день (21 мая 2017 года) Викиданные содержат 131 природоохранную зону по России с заполненным свойством "дата основания"(рис. 4).

[1] Protected area. Research in programming Wikidata [Programmirovaniye Vikidannykh] // Russian Wikiversity, https://ru.wikiversity.org/wiki/%D0%9F%D1%80%D0%BE%D0%B3%D1%80%D0%B0%D0%BC%D0%BC%D0%B8%D1%80%D0%BE%D0%B2%D0%B0%D0%BD%D0%B8%D0%B5_%D0%92%D0%B8%D0%BA%D0%B8%D0%B4%D0%B0%D0%BD%D0%BD%D1%8B%D1%85/%D0%9F%D1%80%D0%B8%D1%80%D0%BE%D0%B4%D0%BE%D0%BE%D1%85%D1%80%D0%B0%D0%BD%D0%BD%D0%B0%D1%8F_%D0%B7%D0%BE%D0%BD%D0%B0, accessed 23 Feb 2017 (2017).

- [2] BioDat, В.В.Дёжкин. Особо охраняемые природные территории мира и некоторые проблемы российского заповедного дела, <http://biodat.ru/doc/lib/degkin2.htm>, accessed 21 May 2017.
- [3] Protected Planet, <http://www.wdpa.org/>, accessed 21 May 2017 (2016).
- [4] I. W. P. Congress, What is protected area?, http://www.worldparkscongress.org/about/what_are_protected_areas.html, accessed 13 Mar 2017 (2014).